
       IDENTIFICATION DIVISION. 
       PROGRAM-ID.           OPPSCAL. 
      *AUTHOR.            ED FRANEY. 
      *REMARKS.                HCFA. 
      *********************************************************** 
      * 5/8/00  - ADD WINX TO RETURN RECORD                     * 
      * 5/10/00 - ADD MSA TO RETURN RECORD                      * 
      * 5/11/00 - CHANGE PROVIDER FILE FROM 9999 OCCURS TO 999  * 
      *         - CHANGE ALL LINE OCCURS FROM 999 TO 450        * 
      * 5/17/00 - RELOCATE MOVE OF H-WINX1 TO A-WINX TO         * 
      *           0000-PROCESS-MAINLINE                         * 
      * 6/08/00 - ADD CODE IN 0550-CALC-STANDARD:               * 
      *              MOVE W-MIN-COIN (W-LP-INDX) TO H-MIN-COIN  * 
      *         - CORRECTED PROBLEM OF COINSURANCE ELECTION     * 
      *           BEING DROPPED AFTER FIRST LINE WAS PROCESSED  * 
      * 6/14/00 - UPDATED WAGE INDEX TABLE                      * 
      *         - CREATED NEW COPY BOOK                         * 
      * 7/10/00 - CORRECTED 0150-INIT PARAGRAPH                 * 
      *         - WILL INCLUDE ALL PACKAGED LINES IN TOTAL      * 
      *           CHARGES FOR OUTLIER CALCULATION               * 
      *           - INCLUDED OPPS-PKG-FLAG = 0 OR 1 OR 2        * 
      * 7/18/00 - CORRECTED MAX-COINSURANCE PARAGRAPH 0550-CALC * 
      *           - CHANGED >>  MOVE 776 TO H-NAT-COIN << TO    * 
      *           -            MOVE H-MAX-COIN TO H-NAT-COIN    * 
      * 8/02/00 - CHANGED OPPSAPCS COPYBOOK TO RANK NEW DEVICES * 
      *           LAST IN DEDUCTIBLE CALCULATION                * 
      *           - OCE SERVICE INDICATOR = H                   * 
      *           - PAYMENT AND COINSURANCE = ZERO              * 
      * 8/02/00 - CHANGED OPPSWINX COPYBOOK TO INCLUDE NEW      * 
      *           MARYLAND CODE                                 * 
      *           - '  80' = 008631 000000                      * 
      * 8/03/00 - CHANGED 0300-COIN-DEDUCT                      * 
      *           - REMOVE CONDITION OF SERVICE INDICATOR       * 
      *             NOT EQUAL 'T'                               * 
      *             - MOVE 1 TO DISCOUNT RATE                   * 
      *           - ALWAYS ACCEPT DISCOUNT FROM OCE             * 
      * 8/07/00 - CHANGED 0150-INIT PARAGRAPH                   * 
      *           - ADD CONDITION OF OPPS-PKG-FLAG NOT = ZERO   * 
      *             - WILL NOT PAY PACKAGED PARTIAL             * 
      *               HOSPITALIZATIONS  FLAG = 1 OR 2           * 
      * 8/07/00 - CHANGED 0250-CALC-DISCOUNT                    * 
      *           - DISCOUNT INDICATOR OF '8' NOW EQUAL TO 2    * 
      *             - FORMULA NOW = 2 (DOUBLE)                  * 
      *             - FORMULA WAS = 2 / UNITS                   * 
      * 8/15/00 - CHANGED 0100-INIT                             * 
      *           - ALLOW PROPER PROCESSING OF SPECIAL WAGE     * 
      *             INDEX CONSIDERATIONS                        * 
      *           - PERFORM 0220-CHNG-WAGEINDX FIRST            * 
      *           - IF WAGE INDEX = 0                           * 
      *              PERFORM 0200-CALC-WAGEINDX                 * 
      * 8/17/00 - CHANGED 0150-INIT                             * 
      *           - ALLOW THE PROCESSING OF PARTIAL HOSP. IF    * 
      *             - LINE ITEM DENIAL/REJECT FLAG = 1          * 
      *               AND APC = 0033,0034,0322-0325,0373,0374   * 
      * 8/18/00 - CHANGED 0900-END-PRICE-RTN                    * 
      *           - PREVENT OUTLIER PROCESSING OF NON-OPPS      * 
      *             CLAIMS                                      * 



      *             - IF TOTAL CLAIM PAYMENT = 0                * 
      *                DO NOT CALCULATE OUTLIER AMOUNT          * 
      * 8/18/00 - ADD 0125-INIT PARAGRAPH                       * 
      *           - SET FLAG IF APC = 0033 ON CLAIM             * 
      * 8/18/00 - CHANGE 0150-INIT                              * 
      *           - IF TYPE OF BILL INCLUSION = 0  (OLD)        * 
      *           - (NEW) OR (APC 0033 IS ON THE CLAIM          * 
      *                   AND SERVICE INDICATOR = 'P'           * 
      *                   OR  APC = 0322-0325,0373,0374)        * 
      * 8/22/00 - CHANGE 0400-CALCULATE                         * 
      *           - IF PROVIDER REDUCED COINSURANCE IS GREATER  * 
      *             THAN NATIONAL ADJUSTED COINSURANCE          * 
      *                 - MOVE NATIONAL COINSURANCE TO THE      * 
      *                   PROVIDER REDUCE COINSURANCE           * 
      * 9/14/00 - CHANGE 0550-CALCULATE-STANDARD                * 
      *           - CHANGE MINIMUM COINSURANCE CALCULATION      * 
      *             FOR DRUGS AND DEVICES (SERVICE INDICATORS   * 
      *             'G','H', OR 'J')                            * 
      * 9/14/00 - CHANGE APC AND WAGE INDEX LOOKUP ROUTINES.    * 
      *           - LOGIC DID NOT ALLOW FOR MULTIPLE RELEASE    * 
      *             DATES                                       * 
      * 10/12/00- CONTROL THE PROCESSING OF PARAGRAPH 0220-     * 
      *             CHNG-WAGEINDX BY USING THE SERVICE FROM     * 
      *             DATE - ONLY PROCESS IF < 20010101          * 
      * 10/12/00- CONTROL THE PROCESSING OF PARAGRAPHS 0105     * 
      *             AND 0110  BY USING THE SERVICE FROM         * 
      *             DATE - RESET FLOOR MSA                      * 
      * 10/12/00- CONTROL H-MAX-COIN USING THE SERVICE FROM     * 
      *             DATE - IP MAX = 776 IF < 20010101           * 
      *                  - IP MAX = 792 IF > 20001231           * 
      *                  (PARAGRAPH 0550)                       * 
      * 10/12/00- LIMIT THE H-MAX-COIN FOR STATUS INDICATORS    * 
      *           'G' 'J' 'K' (DRUGS) TO $792 PER LINE ITEM     * 
      *            EFFECT 20010101 (PARAGRAPH 0550)             * 
      * 10/12/00- ALLOW FOR A NEW SERVICE INDICATOR  'K'        * 
      * 11/15/00- ADJUST THE COST-TO-CHARGE FOR 2001 BY A       * 
      *           FACTOR OF .981956                             * 
      *               - PARAGRAPHS 0555 AND 0910                * 
      * 11/16/00- CREATED WAGE INDEX COPY BOOK FOR 20010101     * 
      * 12/06/00- ALLOW FOR NEW SERVICE INDICATOR  'B'          * 
      *         - NON-ALLOWED ITEM OR SERVICE FOR OPPS          * 
      * 12/07/00- INSERTED W-APC-ADJ-TABLE TO ADJUST APC        * 
      *           FROM OCE                                      * 
      *         - RESET SERVICE INDICATOR IF NECESSARY          * 
      *         - RESET PAYMENT INDICATOR IF NECESSARY          * 
      * 12/07/00- INSERTED PARAGRAPH 0160-ADJUST-APC            * 
      * 12/28/00- CORRECTED WAGE INDEX LOOK-UP ROUTINE          * 
      *         - PARAGRAPH 0210-WAGE-LOOKUP ( > INSTEAD OF < ) * 
      *         - CHECK RECLASS VALUE OF 'Y' ELSE ALLOW ANY     * 
      *           OTHER VALUE FOR NON-RECLASS                   * 
      * 02/28/01- ADDED NEW APCS - 'C' CODES DEVICE PASSTHRUS   * 
      *         - 125 CODES                                     * 
      * 02/28/01- REMOVED OCE/APC PATCH - EFFECTIVE APR 01,2001 * 
      *           (0160-ADJUST-APC)                             * 
      * 03/05/01- SET RETURN CODE TO '30' TO BY-PASS THE PASS   * 
      *           THRU PAYMENTS FOR THE FOLLOWING APCS:         * 
      *           01111 - 01114, 01117, 06300, AND 06600        * 



      * 03/06/01- ADD PATCH FOR HCPCS CODE C1050                * 
      *           CHANGE SERVICE INDICATOR TO 'S'               * 
      *           CHANGE PAYMENT APC TO '0976'                  * 
      * 03/07/01- ADDED NEW APCS - 'C' CODES DEVICE PASSTHRUS   * 
      *         - 25 CODES                                      * 
      * 05/02/01- ADD CODE FOR DAILY COINSURANCE LIMITATION     * 
      *         - UPDATE APC TABLE FOR 7/1/2001                 * 
      *         - NEW PROCESS FOR "DELETED" APC CODES           * 
      * 06/12/01- CORRECT TRUNCATION OF APC RANKING FACTOR      * 
      *         - INCREASE THE SIZE OF THE RANKING VARIABLE     * 
      *         - 9(03) TO 9(05)                                * 
      * 07/23/01- ADD NEW PROVIDER SPECIFIC WAGE INDEX LOGIC    * 
      *         - PARAGRAPH 0225-CHNG-WAGEINDX                  * 
      * 07/23/01- UPDATE APC TABLE                              * 
      * 08/20/01- REMOVE REFERENCE TO L-PSF-GEO-MSA FROM LOGIC  * 
      *         - IN PARAGRAPH 0225-CHNG-WAGEINDX               * 
      *********************************************************** 
       DATE-COMPILED. 
       ENVIRONMENT DIVISION. 
       CONFIGURATION SECTION. 
       SOURCE-COMPUTER.            IBM-370. 
       OBJECT-COMPUTER.            IBM-370. 
       INPUT-OUTPUT  SECTION. 
       FILE-CONTROL. 
 
       DATA DIVISION. 
       FILE SECTION. 
 
       WORKING-STORAGE SECTION. 
       01  W-STORAGE-REF                  PIC X(46)  VALUE 
           'OPCAL2001.3   - W O R K I N G   S T O R A G E'. 
       01  CAL-VERSION                    PIC X(07)  VALUE 'C2001.3'. 
       01  R1                             PIC S9(04) COMP SYNC. 
       01  R2                             PIC S9(04) COMP SYNC. 
       01  R3                             PIC S9(04) COMP SYNC. 
       01  R4                             PIC S9(04) COMP SYNC. 
 
      *************************************************************** 
      *    LAYUP TABLE AREA FOR CY2001 APC ADJUSTMENTS              * 
      *      - DISABLED AS OF 04-01-2001                            * 
      *************************************************************** 
      *01  W-APC-ADJ-FILLS. 
      *    03                             PIC X(18) VALUE 
      *      'G01750060300602V10'. 
      *    03                             PIC X(36) VALUE 
      *      'C85130099600991X10929960008200083T10'. 
      *    03                             PIC X(36) VALUE 
      *      '929950008200083T10929840008300082T10'. 
      *    03                             PIC X(36) VALUE 
      *      '929820008300082T10355000008100091T10'. 
      *    03                             PIC X(36) VALUE 
      *      '194990002900004T10193960002900028T10'. 
      *    03                             PIC X(36) VALUE 
      *      '193800003000029T10193710003000029T10'. 
      *    03                             PIC X(36) VALUE 
      *      '193700003000028T10193660003000029T10'. 
      *    03                             PIC X(36) VALUE 



      *      '193550003000028T10193500003000028T10'. 
      *    03                             PIC X(36) VALUE 
      *      '193300003000028T10193280003000028T10'. 
      *    03                             PIC X(36) VALUE 
      *      '193240003000029T10193180003000029T10'. 
      *    03                             PIC X(36) VALUE 
      *      '193160003000029T10192910002900028T10'. 
      *    03                             PIC X(36) VALUE 
      *      '192900002900028T10191820003000028T10'. 
      *    03                             PIC X(36) VALUE 
      *      '191800003000029T10191620003000029T10'. 
      *    03                             PIC X(36) VALUE 
      *      'J33650703607036K10J29960091500915K10'. 
      *    03                             PIC X(36) VALUE 
      *      'J22600700707007K10J01500091700917K10'. 
      *    03                             PIC X(36) VALUE 
      *      'J15700090700907K10J12450091700917K10'. 
      *    03                             PIC X(36) VALUE 
      *      'C13050130501305G51C10960109601096G51'. 
      * 
      *01  W-APC-ADJ-TABLE REDEFINES W-APC-ADJ-FILLS. 
      *    03  WAAJ-ENTRY OCCURS 33 TIMES 
      *             INDEXED BY WAAJ-INDX. 
      *       10  WAAJ-HCPCSAPC           PIC X(10). 
      *       10  WAAJ-FED-APC            PIC X(05). 
      *       10  WAAJ-NEW-SRVC-IND       PIC X(01). 
      *       10  WAAJ-NEW-PYMT-IND       PIC X(01). 
      *       10  WAAJ-NEW-PYMT-ADJ       PIC X(01). 
      * 
      *01  WAAJ-MAX                       PIC S9(07) COMP-3 
      *                          VALUE +33. 
      *************************************************************** 
 
      *************************************************************** 
      *    LAYUP TABLE AREA FOR CY2000 APC'S                        * 
      *************************************************************** 
      *************************************************************** 
      ****** WAA-INDX   ****************************** 
       COPY OPPSAPCS. 
 
      *************************************************************** 
      *                   THIS IS THE WAGE-INDEX                    * 
      *          ASSOCIATED WITH THE BILL BEING PROCESSED           * 
      *************************************************************** 
      ****** WWM-INDX   ****************************** 
       COPY OPPSWINX. 
 
      *************************************************************** 
      *                   THIS IS THE MAX-COINSURANCE               * 
      *          DETERMINED BY DATE LINE-ITEM PROCESSING            * 
      *************************************************************** 
      ****** WMC-INDX   ****************************** 
 
       01  W-MAX-COIN-DATE-FILLS. 
           03                          PIC X(22)  VALUE 
              '2000080100120010101002'. 
       01  W-MAX-COIN-DATE-TABLE REDEFINES W-MAX-COIN-DATE-FILLS. 



           03  WMC-ENTRY  OCCURS 2 TIMES. 
              05  WMC-DATE             PIC X(8). 
              05  WMC-DTCD             PIC 9(3). 
 
       01  WORK-AREA. 
           05  H-SUB                   PIC S9(07) COMP-3  VALUE ZERO. 
           05  W-SUB1                  PIC S9(07) COMP-3  VALUE ZERO. 
           05  W-SUB2                  PIC S9(07) COMP-3  VALUE ZERO. 
           05  PS-SUB                  PIC S9(07) COMP-3  VALUE ZERO. 
           05  LN-SUB                  PIC S9(07) COMP-3  VALUE ZERO. 
           05  DISC-FRACTION           PIC 9V9(03)  VALUE .500. 
           05  TERM-PROC-DISC          PIC 9V9(03)  VALUE .500. 
           05  APC33-FLAG              PIC X(01). 
           05  GJK-FLAG                PIC X(01). 
           05  T-LITEM-PYMT            PIC S9(07)V9(02). 
           05  W-ADJ-HCPCSAPC. 
               10  W-ADJ-HCPCS         PIC X(05). 
               10  W-ADJ-APC           PIC X(05). 
 
      **************************************************************** 
      *   BELOW ARE THE VARIABLES THAT WILL BE HELD FOR CLAIM        * 
      *   LEVEL PROCESSING (OUTLIER/DEDUCTIBLE/COINSURANCE)          * 
      **************************************************************** 
       01  H-ADDITIONAL-VARIABLES. 
           05  H-OUTLIER-PYMT             PIC S9(07)V9(02). 
           05  H-PRTL-HOSP-PYMT           PIC 9(07)V9(02). 
           05  H-TOTAL-CLM-DEDUCT         PIC 9(03)V9(02). 
           05  H-TOT-CHRG                 PIC 9(07)V9(02). 
           05  H-TOT-PYMT                 PIC 9(07)V9(02). 
           05  H-BENE-DEDUCT              PIC 9(03)V9(02). 
           05  H-SUB-CHRG                 PIC 9(08)V9(02). 
           05  H-MAX-COIN                 PIC 9(05)V9(02). 
           05  H-IP-LIMIT                 PIC 9(05)V9(02). 
           05  H-NEW-COIN                 PIC 9(05)V9(02). 
           05  H-NEW-WGNAT                PIC 9(05)V9(02). 
           05  LINE-HOLD-ITEMS. 
              10  H-COIN-PERCENT          PIC 9(01)V9(04). 
              10  H-LITEM-PYMT            PIC 9(07)V9(02). 
              10  H-LITEM-REIM            PIC 9(07)V9(02). 
              10  H-APC-PYMT              PIC 9(07)V9(02). 
              10  H-TOTAL-LN-DEDUCT       PIC 9(03)V9(02). 
              10  H-NAT-COIN              PIC 9(05)V9(02). 
              10  H-MIN-COIN              PIC 9(05)V9(02). 
              10  H-PSF-COIN              PIC 9(05)V9(02). 
              10  H-RED-COIN              PIC 9(05)V9(02). 
              10  H-RATIO                 PIC S9(03)V9(07). 
              10  H-SHIFT                 PIC 9(05)V9(02). 
              10  H-TOTAL                 PIC 9(05)V9(02). 
              10  H-LN-PTR                PIC 9(03). 
              10  H-SRVC-UNITS            PIC 9(07). 
              10  H-RANK                  PIC 9(05). 
              10  H-PSF-MSA               PIC X(04). 
              10  H-DCP-STAGE. 
                15  H-DCP-DOS             PIC 9(08). 
                15  H-DCP-CODE            PIC 9(01). 
              10  H-PPCT                  PIC S9V9(06) COMP-3. 
              10  H-DISC-RATE             PIC S9V9(08) COMP-3. 



              10  H-WINX1                 PIC S9V9(04) COMP-3. 
 
      **************************************************************** 
      *   BELOW IS THE LAY-UP TABLE TO PROCESS DEDUCTIBLES.          * 
      *                                                              * 
      **************************************************************** 
      ****** W-LP-INDX  ****************************** 
 
       01  W-LNC-MAX                      PIC S9(07)  COMP-3 VALUE +0. 
       01  W-LINE-PTR-TABLE. 
          05  W-LP-ENTRY 
                OCCURS 0 TO 450 TIMES 
                DEPENDING ON W-LNC-MAX 
                ASCENDING KEY IS W-RANK 
                INDEXED BY W-LP-INDX. 
              10  W-LP-SUB                PIC S9(07)   COMP-3. 
              10  W-APC-PYMT              PIC 9(07)V99. 
              10  W-NAT-COIN              PIC 9(05)V99. 
              10  W-MIN-COIN              PIC 9(05)V99. 
              10  W-RED-COIN              PIC 9(04)V99. 
              10  W-DISC-RATE             PIC 9(01)V9(08). 
              10  W-SRVC-UNITS            PIC 9(07). 
              10  W-RANK                  PIC 9(05). 
              10  W-PPCT                  PIC S9V9(06) COMP-3. 
              10  W-WINX1                 PIC S9V9(04) COMP-3. 
 
 
      **************************************************************** 
      *   BELOW IS THE LAY-UP TABLE TO PROCESS DRUG COINSURANCE      * 
      *      ROLL-UP                                                 * 
      **************************************************************** 
      ****** W-DCP-INDX  ****************************** 
 
       01  W-DCP-MAX                      PIC S9(07)  COMP-3 VALUE +0. 
       01  W-DOS-COIN-PTR-TABLE. 
          05  W-DCP-ENTRY 
                OCCURS 0 TO 450 TIMES 
                DEPENDING ON W-DCP-MAX 
                ASCENDING KEY IS W-DCP-STAGE 
                INDEXED BY W-DCP-INDX. 
              10  W-DCP-SUB               PIC S9(07). 
              10  W-DCP-STAGE. 
                 15  W-DCP-DOS            PIC 9(08). 
                 15  W-DCP-CODE           PIC 9(01). 
              10  W-DCP-SRVC-IND          PIC X(01). 
              10  W-DCP-COIN1             PIC 9(05)V99. 
              10  W-DCP-COIN2             PIC 9(05)V99. 
              10  W-DCP-WGNAT             PIC 9(05)V99. 
 
       LINKAGE SECTION. 
      *************************************************************** 
      *    LAYUP TABLE AREA FOR CY2000 PROVIDER SPECIFIC RECORD     * 
      *************************************************************** 
 
       01  L-PROV-SPEC-AREA. 
               05  L-PSF-NPI                    PIC X(08). 
               05  L-PSF-NPI-FILLER             PIC X(02). 



               05  L-PSF-PROV-OSCAR. 
                 10  L-PSF-PROV-ST              PIC X(02). 
                 10  FILLER                     PIC X(04). 
               05  L-PSF-EFFDT                  PIC 9(08). 
               05  L-PSF-FY-BEGIN-DT            PIC 9(08). 
               05  L-PSF-REPORT-DT              PIC 9(08). 
               05  L-PSF-TERMDT                 PIC 9(08). 
               05  L-PSF-WAIVE-IND              PIC X(01). 
               05  L-PSF-FI-NUM                 PIC 9(05). 
               05  L-PSF-PROV-TYPE              PIC X(02). 
               05  FILLER                       PIC X(01). 
               05  L-PSF-WGIDX-RECLASS          PIC X(01). 
               05  L-PSF-GEO-MSA                PIC X(04). 
               05  L-PSF-WI-MSA                 PIC X(04). 
               05  L-PSF-COLA                   PIC 9V9(03). 
               05  L-PSF-BED-SIZE               PIC 9(05). 
               05  L-PSF-OPCOST-RATIO           PIC 9V9(03). 
               05  FILLER                       PIC X(17). 
               05  L-PSF-APC-LINE-CNT           PIC 9(04). 
               05  L-PSF-APC-TABLE     OCCURS 999 TIMES 
                        DEPENDING ON L-PSF-APC-LINE-CNT. 
                   10  L-PSF-APC                PIC X(04). 
                   10  L-PSF-RED-COIN           PIC 9(04)V99. 
 
      *************************************************************** 
      *  INPUT RECORD FROM THE OCE/STANDARD SYSTEM                  * 
      *                                                             * 
      *************************************************************** 
 
       01  OPPS-LINE-CNT                  PIC 9(08) COMP. 
       01  OCE-DATA. 
           05  OPPS-OCE-LINE OCCURS 450 TIMES 
                     DEPENDING ON OPPS-LINE-CNT. 
               10  OPPS-HCPCS             PIC X(05). 
               10  OPPS-GRP. 
                 15  FILLER               PIC X(01). 
                 15  OPPS-APC             PIC X(04). 
               10  OPPS-HCPCS-APC         PIC X(05). 
               10  OPPS-SRVC-IND          PIC X(01). 
               10  OPPS-PYMT-IND          PIC X(01). 
               10  OPPS-DISC-FACT         PIC 9(01). 
               10  OPPS-LITEM-DR-FLAG     PIC X(01). 
               10  OPPS-PKG-FLAG          PIC X(01). 
               10  OPPS-PYMT-ADJ-FLAG     PIC X(01). 
               10  OPPS-SITE-SRVC-FLAG    PIC X(01). 
               10  OPPS-SRVC-UNITS        PIC 9(07). 
               10  OPPS-SUB-CHRG          PIC 9(08)V99. 
               10  OPPS-LITEM-ACT-FLAG    PIC X(01). 
       01  L-SERVICE-FROM-DATE       PIC 9(08). 
       01  BENE-DEDUCT               PIC 9(03)V9(02). 
      *************************************************************** 
      *   BELOW ARE THE VARIABLES THAT WILL BE PASSED BACK          * 
      *    TO SS ASSOCIATED WITH THE BILL BEING PROCESSED           * 
      *************************************************************** 
       01  A-ADDITIONAL-VARIABLES. 
           05  A-CALC-VERS             PIC X(07). 
           05  A-TOTAL-CLM-DEDUCT      PIC 9(03)V9(02). 



           05  A-OUTLIER-PYMT          PIC 9(07)V9(02). 
           05  A-TOT-CLM-PYMT          PIC 9(07)V9(02). 
           05  A-TOT-CLM-CHRG          PIC 9(07)V9(02). 
           05  A-CLM-RTN-CODE          PIC 9(02). 
           05  A-MSA                   PIC X(04). 
           05  A-WINX                  PIC S9V9(04). 
           05  A-LINE-ITEMS OCCURS 450 TIMES 
                   DEPENDING ON OPPS-LINE-CNT. 
             10  A-LITEM-PYMT          PIC 9(07)V9(02). 
             10  A-LITEM-REIM          PIC 9(07)V9(02). 
             10  A-TOTAL-LN-DEDUCT     PIC 9(03)V9(02). 
             10  A-ADJ-COIN            PIC 9(05)V9(02). 
             10  A-RED-COIN            PIC 9(05)V9(02). 
             10  A-RETURN-CODE         PIC 9(02). 
 
       01  OCE-IN-DATE. 
           05  OCE-IN-LINES OCCURS 450 TIMES. 
              10  FILLER               PIC X(15). 
              10  OPPS-LITEM-DOS       PIC 9(08). 
              10  FILLER               PIC X(21). 
      *************************************************************** 
      *    PROCESSING:                                              * 
      *        A. WILL PROCESS LINE ITEMS BASED ON OCE FLAGS.       * 
      *        B. INITIALIZE OPPS   HOLD VARIABLES.                 * 
      *        C. EDIT THE DATA PASSED FROM THE OCE.                * 
      *        D. ASSEMBLE PRICING COMPONENTS.                      * 
      *        E. CALCULATE THE PRICE.                              * 
      *        F. RETURN COINSURANCE/REIMBURSEMENT/DEDUCTIBLE/      * 
      *                  PAYMENT/OUTLIER AMOUNT/RETURN CODES        * 
      *************************************************************** 
 
       PROCEDURE DIVISION  USING OPPS-LINE-CNT 
                                 OCE-DATA 
                                 A-ADDITIONAL-VARIABLES 
                                 L-PROV-SPEC-AREA 
                                 L-SERVICE-FROM-DATE 
                                 BENE-DEDUCT 
                                 OCE-IN-DATE. 
 
          0000-PROCESS-MAINLINE. 
 
              PERFORM 0100-INIT 
                 THRU 0100-INIT-EXIT. 
              IF A-CLM-RTN-CODE >= 50 
                 GOBACK. 
              MOVE H-WINX1 TO A-WINX. 
              PERFORM 0125-INIT 
                 THRU 0125-INIT-EXIT 
                   VARYING LN-SUB FROM 1 BY 1 
                     UNTIL LN-SUB > OPPS-LINE-CNT. 
              PERFORM 0150-INIT 
                 THRU 0150-INIT-EXIT 
                   VARYING LN-SUB FROM 1 BY 1 
                     UNTIL LN-SUB > OPPS-LINE-CNT. 
              MOVE 0 TO W-DCP-MAX. 
              PERFORM 0400-CALCULATE 
                 THRU 0400-CALCULATE-EXIT 



                    VARYING W-LP-INDX FROM 1 BY 1 
                      UNTIL W-LP-INDX > W-LNC-MAX. 
              IF GJK-FLAG = 'Y' 
                 PERFORM 0800-ADJ-STV-REIM 
                    THRU 0800-ADJ-STV-REIM-EXIT 
                       VARYING W-DCP-INDX FROM 1 BY 1 
                         UNTIL W-DCP-INDX > W-DCP-MAX 
              ELSE 
                 NEXT SENTENCE. 
              PERFORM 0900-END-PRICE-RTN 
                 THRU 0900-END-PRICE-RTN-EXIT. 
 
           GOBACK. 
 
      *************************************************************** 
      *    INITIALIZE WORKING STORAGE HOLD AREAS                    * 
      *    AND ADDITIONAL VARIABLES TO BE PASSED BACK TO            * 
      *    THE STANDARD SYSTEM.                                     * 
      *    - IF PROVIDER SPECIFIC FILE WAGE INDEX RECLASSIFICATION  * 
      *      CODE INVALID OR MISSING                                * 
      *       - MOVE '52' TO CLAIM LEVEL RETURN CODE                * 
      *       - DISCONTINUE CLAIM PROCESSING                        * 
      *    - IF SERVICE FROM DATE NOT NUMERIC OR < 20000801         * 
      *       - MOVE '53' TO CLAIM LEVEL RETURN CODE                * 
      *       - DISCONTINUE CLAIM PROCESSING                        * 
      *    - IF SERVICE FROM DATE < PROVIDER EFFECTIVE DATE         * 
      *       - MOVE '54' TO CLAIM LEVEL RETURN CODE                * 
      *       - DISCONTINUE CLAIM PROCESSING                        * 
      *    - IF SERVICE FROM DATE > PROVIDER TERMINATION DATE       * 
      *       - MOVE '54' TO CLAIM LEVEL RETURN CODE                * 
      *       - DISCONTINUE CLAIM PROCESSING                        * 
      *************************************************************** 
         0100-INIT. 
 
             MOVE 01 TO A-CLM-RTN-CODE. 
             MOVE 'N' TO APC33-FLAG GJK-FLAG. 
             IF L-SERVICE-FROM-DATE NOT NUMERIC 
                MOVE 53 TO A-CLM-RTN-CODE 
                GO TO 0100-INIT-EXIT 
             ELSE 
                IF L-SERVICE-FROM-DATE < 20000801 
                   MOVE 53 TO A-CLM-RTN-CODE 
                   GO TO 0100-INIT-EXIT 
                ELSE 
                   IF L-SERVICE-FROM-DATE < L-PSF-EFFDT 
                      MOVE 54 TO A-CLM-RTN-CODE 
                      GO TO 0100-INIT-EXIT 
                   ELSE 
                      IF L-PSF-TERMDT > 0 
                         IF L-SERVICE-FROM-DATE > L-PSF-TERMDT 
                            MOVE 54 TO A-CLM-RTN-CODE 
                            GO TO 0100-INIT-EXIT 
                         END-IF 
                      END-IF. 
             MOVE SPACE TO A-MSA. 
             MOVE ZERO TO A-OUTLIER-PYMT 
                          A-TOTAL-CLM-DEDUCT 



                          A-TOT-CLM-CHRG 
                          A-TOT-CLM-PYMT 
                          A-WINX 
                          W-LNC-MAX. 
             INITIALIZE H-ADDITIONAL-VARIABLES. 
             INITIALIZE LINE-HOLD-ITEMS. 
             INITIALIZE T-LITEM-PYMT. 
             INITIALIZE W-ADJ-HCPCSAPC. 
             MOVE CAL-VERSION TO A-CALC-VERS. 
             MOVE BENE-DEDUCT TO H-BENE-DEDUCT. 
             MOVE WAD-MAX TO WAD-SUB. 
             PERFORM UNTIL L-SERVICE-FROM-DATE 
                     NOT < WAD-DATE (WAD-SUB) 
                 SUBTRACT 1 FROM WAD-SUB 
             END-PERFORM. 
             IF L-PSF-WGIDX-RECLASS = 'Y' 
                MOVE L-PSF-WI-MSA TO H-PSF-MSA 
             ELSE 
                IF L-PSF-WGIDX-RECLASS = 'N' 
                   MOVE L-PSF-GEO-MSA TO H-PSF-MSA 
                ELSE 
                   MOVE  52  TO A-CLM-RTN-CODE 
                   GO TO 0100-INIT-EXIT. 
 
             IF L-SERVICE-FROM-DATE >= 20010101 
                MOVE 792 TO H-IP-LIMIT 
             ELSE 
               IF L-SERVICE-FROM-DATE >= 20000801 
                  MOVE 776 TO H-IP-LIMIT. 
 
             IF L-SERVICE-FROM-DATE < 20010101 
                PERFORM 0105-FLOOR-2000 
                   THRU 0105-FLOOR-2000-EXIT. 
 
             IF L-SERVICE-FROM-DATE > 20001231 
                PERFORM 0110-FLOOR-2001 
                   THRU 0110-FLOOR-2001-EXIT. 
 
                MOVE H-PSF-MSA TO A-MSA. 
 
             IF L-SERVICE-FROM-DATE < 20010101 
                PERFORM 0220-CHNG-WAGEINDX 
                   THRU 0220-CHNG-WAGEINDX-EXIT 
             ELSE 
               IF L-SERVICE-FROM-DATE < 20020101 
                 PERFORM 0225-CHNG-WAGEINDX 
                    THRU 0225-CHNG-WAGEINDX-EXIT. 
             IF H-WINX1 = 0 THEN 
                PERFORM 0200-CALC-WAGEINDX 
                   THRU 0200-CALC-WAGEINDX-EXIT. 
 
         0100-INIT-EXIT. 
            EXIT. 
 
      *************************************************************** 
      *  RESET FLOOR MSA - 'FROM-DATE' CONTROLLED                   * 
      *     - YEAR 2000                                             * 



      *     - YEAR 2001                                             * 
      *************************************************************** 
 
         0105-FLOOR-2000. 
 
             IF (H-PSF-MSA = '6020' OR '9000' OR '8080') AND 
                (L-PSF-PROV-ST = '36') 
                  MOVE '  36' TO H-PSF-MSA 
             ELSE 
                IF (H-PSF-MSA = '2440') AND 
                   (L-PSF-PROV-ST = '15') 
                     MOVE '  15' TO H-PSF-MSA 
                ELSE 
                   IF (H-PSF-MSA = '2520') AND 
                      (L-PSF-PROV-ST = '24') AND 
                      (L-PSF-WGIDX-RECLASS = 'Y') 
                        MOVE '  24' TO H-PSF-MSA 
                   ELSE 
                      IF (H-PSF-MSA = '1123') AND 
                         (L-PSF-PROV-ST = '22') 
                            MOVE '  22' TO H-PSF-MSA. 
 
 
         0105-FLOOR-2000-EXIT. 
            EXIT. 
 
         0110-FLOOR-2001. 
 
             IF (H-PSF-MSA = '6020' OR '9000' OR '8080') AND 
                (L-PSF-PROV-ST = '36') 
                  MOVE '  36' TO H-PSF-MSA 
             ELSE 
                IF (H-PSF-MSA = '2440') AND 
                   (L-PSF-PROV-ST = '15') 
                     MOVE '  15' TO H-PSF-MSA 
                ELSE 
                   IF (H-PSF-MSA = '9000') AND 
                      (L-PSF-PROV-ST = '51') 
                        MOVE '  51' TO H-PSF-MSA 
                   ELSE 
                      IF (H-PSF-MSA = '1900') AND 
                         (L-PSF-PROV-ST = '21') 
                            MOVE '  21' TO H-PSF-MSA 
                     ELSE 
                        IF (H-PSF-MSA = '1123') AND 
                           (L-PSF-PROV-ST = '22') 
                              MOVE '  22' TO H-PSF-MSA. 
 
         0110-FLOOR-2001-EXIT. 
            EXIT. 
 
      *************************************************************** 
      *  SET FLAG IF APC = 0033                                     * 
      *    - TERMINATE PROCESS IF 0033 LOCATED                      * 
      *                                                             * 
      *************************************************************** 
 



         0125-INIT. 
 
             IF OPPS-APC (LN-SUB) = '0033' 
                MOVE 'Y' TO APC33-FLAG 
                MOVE 451 TO LN-SUB. 
 
         0125-INIT-EXIT. 
            EXIT. 
 
      *************************************************************** 
      * 1. VERIFY THE SERVICE INDICATOR PASSED BY THE OCE           * 
      *      - IF INVALID SET RETURN CODE TO '40'                   * 
      *         - DISCONTINUE LINE PROCESSING                       * 
      *      - IF VALID SET RETURN CODE TO '01'                     * 
      *         - CONTINUE LINE PROCESSING                          * 
      * 2. PROCESS LINES AND LOAD WORK TABLE ACCORDING TO PRICE     * 
      *    RANKING                                                  * 
      * 3. CHECK OCE EDIT INDICATORS AND SET RETURN CODES IF        * 
      *    INDICATORS ARE INVALID.                                  * 
      *     - VALID RETURN CODES FOR EDIT INDICATORS                * 
      *       - '41' - SERVICE INDICATOR INVALID FOR OPPS PRICER    * 
      *       - '42' - APC = '00000' OR (PACKAGING FLAG = 1 OR 2)   * 
      *       - '43' - PAYMENT INDICATOR NOT = TO 1 OR 5 THRU 9     * 
      *       - '44' - SERVICE INDICATOR = 'H' BUT PAYMENT          * 
      *                  INDICATOR NOT = TO 6                       * 
      *       - '45' - PACKAGING FLAG NOT = TO 0                    * 
      *       - '46' - (LINE ITEM DENIAL/REJECT FLAG NOT = TO 0     * 
      *                 OR LINE ITEM DENIAL/REJECT FLAG  = TO 1     * 
      *                 AND (APC NOT = 0033 OR 0034 OR 0322 OR      * 
      *                      0323 OR 0324 OR 0325 OR 0373 OR 0374)) * 
      *                 OR LINE ITEM ACTION FLAG NOT = TO 1         * 
      *       - '47' - LINE ITEM ACTION FLAG = 2 OR 3               * 
      *       - '48' - PAYMENT ADJUSTMENT FLAG NOT VALID            * 
      *       - '49' - SITE OF SERVICE FLAG NOT = TO 0              * 
      *               - OR (APC 0033 IS NOT ON THE CLAIM            * 
      *                     AND SERVICE INDICATOR = 'P'             * 
      *                     OR  APC = 0322-0325,0373,0374)          * 
      * 4. IF OCE INDICATORS ARE VALID, SEARCH APC TABLE            * 
      *     - IF MISSING, DELETED OR INVALID APC                    * 
      *       - SET RETURN CODE TO '30'                             * 
      *         - DISCONTINUE LINE PROCESSING                       * 
      *************************************************************** 
         0150-INIT. 
 
      *************************************************************** 
      *  - DISABLED AS OF 04-01-2001                                * 
      *      PERFORM 0160-ADJUST-APC                                * 
      *         THRU 0160-ADJUST-APC-EXIT.                          * 
      *************************************************************** 
      *  DISABLED AS OF 07-01-2001                                  * 
      *************************************************************** 
      *      IF (L-SERVICE-FROM-DATE > 20010331) AND                * 
      *         (OPPS-APC (LN-SUB) = '1410') AND                    * 
      *         (OPPS-HCPCS (LN-SUB) = 'C1050')                     * 
      *           MOVE 'S' TO  OPPS-SRVC-IND (LN-SUB)               * 
      *           MOVE '0976' TO OPPS-APC (LN-SUB).                 * 
      *************************************************************** 



 
 
             SET W-LP-INDX TO LN-SUB. 
             MOVE ZERO TO W-LP-ENTRY (W-LP-INDX). 
             MOVE ZERO TO A-LITEM-PYMT (LN-SUB) 
                          A-LITEM-REIM (LN-SUB) 
                          A-ADJ-COIN (LN-SUB) 
                          A-RED-COIN (LN-SUB) 
                          A-TOTAL-LN-DEDUCT (LN-SUB). 
 
      *      IF OPPS-APC (LN-SUB) = '1111' OR '1112' OR '1113' 
      *            OR '1114' OR '1117' OR '6300' OR '6600' 
      *        IF L-SERVICE-FROM-DATE > 20010331 
      *           MOVE  30  TO A-RETURN-CODE (LN-SUB) 
      *           GO TO 0150-INIT-EXIT. 
 
             IF NOT (OPPS-SRVC-IND (LN-SUB) = 'A' OR 'B' OR 'C' OR 
              'E' OR 'F' OR 'G' OR 'H' OR 'J' OR 'K' OR 'N' OR 'P' 
               OR 'S' OR 'T' OR 'V' OR 'X') THEN 
                MOVE  40  TO A-RETURN-CODE (LN-SUB) 
                GO TO 0150-INIT-EXIT 
              ELSE 
                MOVE  01  TO A-RETURN-CODE (LN-SUB). 
 
             IF OPPS-SRVC-IND (LN-SUB) = 
                          'A' OR 'B' OR 'C' OR 'E' OR 'F' 
                MOVE  41  TO A-RETURN-CODE (LN-SUB) 
                GO TO 0150-INIT-EXIT. 
 
               IF OPPS-PYMT-IND (LN-SUB) = '1' OR '5' OR '6' OR 
                                           '7' OR '8' OR '9' 
                 IF OPPS-PKG-FLAG (LN-SUB) = '0' OR '1' OR '2' 
                   IF (OPPS-LITEM-DR-FLAG (LN-SUB) = '0' OR 
                       OPPS-LITEM-DR-FLAG (LN-SUB) = '1' AND 
                       (OPPS-APC (LN-SUB) = '0033' OR '0034' 
                       OR '0322' OR '0323' OR '0324' OR '0325' 
                       OR '0373' OR '0374')) OR 
                      (OPPS-LITEM-ACT-FLAG (LN-SUB) = '1') 
                     IF NOT (OPPS-LITEM-ACT-FLAG (LN-SUB) 
                                         EQUAL '2' OR '3') 
                       IF OPPS-PYMT-ADJ-FLAG (LN-SUB) = '0' OR '1' 
                                           OR '2' OR '3' OR '4' 
                         IF (OPPS-SITE-SRVC-FLAG (LN-SUB) = '0') OR 
                            ((APC33-FLAG = 'Y') AND 
                             ((OPPS-SRVC-IND (LN-SUB) = 'P') OR 
                             (OPPS-APC (LN-SUB) = '0322' OR '0323' 
                              OR '0324' OR '0325' OR '0373' 
                              OR '0374'))) 
                MOVE OPPS-SRVC-UNITS (LN-SUB) TO H-SRVC-UNITS 
 
                MOVE OPPS-SUB-CHRG (LN-SUB) TO H-SUB-CHRG 
                COMPUTE H-TOT-CHRG = H-TOT-CHRG + 
                                     H-SUB-CHRG 
                    IF (OPPS-APC (LN-SUB) = '0000') OR 
                       (OPPS-PKG-FLAG (LN-SUB) NOT = 0) 
                       MOVE 42 TO A-RETURN-CODE (LN-SUB) 
                       GO TO 0150-INIT-EXIT 



                    END-IF 
                SEARCH ALL WAA-ENTRY 
                   AT END 
                      MOVE 30 TO A-RETURN-CODE (LN-SUB) 
                      GO TO 0150-INIT-EXIT 
                   WHEN WAA-APC (WAA-INDX) = OPPS-GRP (LN-SUB) 
                      MOVE WAA-PTR (WAA-INDX) TO W-SUB2 
                      PERFORM 0175-APC-LOOKUP 
                         ELSE 
                            MOVE  49  TO A-RETURN-CODE (LN-SUB) 
                            GO TO 0150-INIT-EXIT 
                       ELSE 
                          MOVE  48  TO A-RETURN-CODE (LN-SUB) 
                          GO TO 0150-INIT-EXIT 
                     ELSE 
                        MOVE  47  TO A-RETURN-CODE (LN-SUB) 
                        GO TO 0150-INIT-EXIT 
                   ELSE 
                      MOVE  46  TO A-RETURN-CODE (LN-SUB) 
                      GO TO 0150-INIT-EXIT 
                 ELSE 
                    MOVE  45  TO A-RETURN-CODE (LN-SUB) 
                    GO TO 0150-INIT-EXIT 
               ELSE 
                  MOVE  43  TO A-RETURN-CODE (LN-SUB) 
                  GO TO 0150-INIT-EXIT. 
 
 
             IF A-RETURN-CODE (LN-SUB) = 01 
               PERFORM 0250-CALC-DISCOUNT 
                  THRU 0250-CALC-DISCOUNT-EXIT 
             ELSE 
                GO TO 0150-INIT-EXIT. 
             IF A-RETURN-CODE (LN-SUB) =  01 
                PERFORM 0300-COIN-DEDUCT 
                   THRU 0300-COIN-DEDUCT-EXIT. 
 
         0150-INIT-EXIT. 
            EXIT. 
 
      *************************************************************** 
      *  ADJUST APC FROM 3M USING THE HCPCS CODE AND APC            * 
      *    - RESET SERVICE INDICATOR WHEN NECESSARY                 * 
      *      - DISABLED AS OF 04-01-2001                            * 
      *************************************************************** 
      *  0160-ADJUST-APC. 
      * 
      *      MOVE OPPS-HCPCS (LN-SUB) TO W-ADJ-HCPCS. 
      *      MOVE OPPS-GRP (LN-SUB) TO W-ADJ-APC. 
      *      SET WAAJ-INDX TO 1. 
      *      SEARCH WAAJ-ENTRY VARYING WAAJ-INDX 
      *         AT END 
      *            GO TO 0160-ADJUST-APC-EXIT 
      *         WHEN WAAJ-HCPCSAPC (WAAJ-INDX) = W-ADJ-HCPCSAPC 
      *            MOVE WAAJ-FED-APC (WAAJ-INDX) TO 
      *                           OPPS-GRP (LN-SUB) 
      *            MOVE WAAJ-NEW-SRVC-IND (WAAJ-INDX) TO 



      *                           OPPS-SRVC-IND (LN-SUB) 
      *            MOVE WAAJ-NEW-PYMT-IND (WAAJ-INDX) TO 
      *                           OPPS-PYMT-IND (LN-SUB) 
      *            MOVE WAAJ-NEW-PYMT-ADJ (WAAJ-INDX) TO 
      *                           OPPS-PYMT-ADJ-FLAG (LN-SUB). 
      *  0160-ADJUST-APC-EXIT. 
      *     EXIT. 
      *************************************************************** 
 
         0175-APC-LOOKUP. 
 
             IF WAR-DTCD (W-SUB2) NOT > WAD-DTCD (WAD-SUB) 
               IF WAR-RATEX (W-SUB2) = 'DELETED' 
                 MOVE 30 TO A-RETURN-CODE (LN-SUB) 
                 COMPUTE H-TOT-CHRG = H-TOT-CHRG - 
                                      H-SUB-CHRG 
               ELSE 
                 MOVE WAR-RATE (W-SUB2) TO H-APC-PYMT 
                 MOVE WAR-RANK (W-SUB2) TO H-RANK 
                 MOVE WAR-MINC (W-SUB2) TO H-MIN-COIN 
                 MOVE WAR-COIN (W-SUB2) TO H-NAT-COIN 
                 MOVE WAR-PPCT (W-SUB2) TO H-PPCT 
             ELSE 
                SUBTRACT 1 FROM W-SUB2 
                IF W-SUB2 > WAA-PTR (WAA-INDX - 1) 
                  GO TO 0175-APC-LOOKUP 
                ELSE 
                   MOVE 0 TO H-APC-PYMT 
                             H-RANK 
                             H-MIN-COIN 
                             H-NAT-COIN 
                             H-PPCT. 
 
         0175-APC-LOOKUP-EXIT. 
            EXIT. 
 
      *************************************************************** 
      *    SEARCH WAGE INDEX TABLE AND SELECT THE WAGE INDEX        * 
      *    WHEN EQUAL TO PROVIDER SPECIFIC WAGE INDEX               * 
      *    IF WAGE INDEX NOT LOCATED                                * 
      *       - SET CLAIM RETURN CODE TO '50'                       * 
      *       - DISCONTINUE CLAIM PROCESSING                        * 
      *    IF WAGE INDEX EQUALS ZERO                                * 
      *       - SET CLAIM RETURN CODE TO '51'                       * 
      *       - DISCONTINUE CLAIM PROCESSING                        * 
      *************************************************************** 
         0200-CALC-WAGEINDX. 
 
             MOVE WWD-MAX TO WWD-SUB. 
             PERFORM UNTIL L-SERVICE-FROM-DATE NOT < WWD-DATE (WWD-SUB) 
                 SUBTRACT 1 FROM WWD-SUB 
             END-PERFORM. 
 
             SEARCH ALL WWM-ENTRY 
                AT END 
                   MOVE  50  TO A-CLM-RTN-CODE 
                   GO TO 0200-CALC-WAGEINDX-EXIT 



                WHEN WWM-MSA (WWM-INDX) = H-PSF-MSA 
                  MOVE WWM-PTR (WWM-INDX) TO W-SUB2 
                  PERFORM 0210-WAGE-LOOKUP. 
 
             IF H-WINX1 = 0 THEN 
                MOVE  51  TO A-CLM-RTN-CODE. 
 
         0200-CALC-WAGEINDX-EXIT. 
             EXIT. 
 
         0210-WAGE-LOOKUP. 
 
             IF WWW-DTCD (W-SUB2) NOT > WWD-DTCD (WWD-SUB) 
                IF L-PSF-WGIDX-RECLASS = 'Y' 
                  MOVE WWW-WINX2 (W-SUB2) TO H-WINX1 
                ELSE 
                  MOVE WWW-WINX1 (W-SUB2) TO H-WINX1 
                END-IF 
             ELSE 
                SUBTRACT 1 FROM W-SUB2 
                IF W-SUB2 > WWM-PTR (WWM-INDX - 1) 
                   GO TO 0210-WAGE-LOOKUP 
                ELSE 
                   MOVE 0 TO H-WINX1. 
 
         0210-WAGE-LOOKUP-EXIT. 
             EXIT. 
 
         0220-CHNG-WAGEINDX. 
 
             IF (L-PSF-PROV-OSCAR = '140012' OR '150002' OR '150004' 
                                OR '150008' OR '150034' OR '150090' 
                                OR '150125' OR '150128' OR '150132') 
                     AND (L-PSF-GEO-MSA = '1600' 
                     AND L-PSF-WI-MSA = '1600' 
                     AND L-PSF-WGIDX-RECLASS = 'Y') 
                MOVE 01.0750 TO H-WINX1. 
 
             IF (L-PSF-PROV-OSCAR = '250078') 
                     AND (L-PSF-GEO-MSA = '3285' 
                     AND L-PSF-WI-MSA = '3285' 
                     AND L-PSF-WGIDX-RECLASS = 'Y') 
                MOVE 00.7634 TO H-WINX1. 
 
             IF (L-PSF-PROV-OSCAR = '330001' OR '330126' OR '330135' 
                                OR '330205' OR '330209' OR '330264') 
                     AND (L-PSF-GEO-MSA = '5600' 
                     AND L-PSF-WI-MSA = '5600' 
                     AND L-PSF-WGIDX-RECLASS = 'Y') 
                MOVE 01.4342 TO H-WINX1. 
 
             IF (L-PSF-PROV-OSCAR = '340039' OR '340129' OR '340144') 
                     AND (L-PSF-GEO-MSA = '1520' 
                     AND L-PSF-WI-MSA = '1520' 
                     AND L-PSF-WGIDX-RECLASS = 'Y') 
                MOVE 00.9434 TO H-WINX1. 
 



             IF (L-PSF-PROV-OSCAR = '360046' OR '360056' OR '360076' 
                                OR '360132') 
                     AND (L-PSF-GEO-MSA = '1640' 
                     AND L-PSF-WI-MSA = '1640' 
                     AND L-PSF-WGIDX-RECLASS = 'Y') 
                MOVE 00.9419 TO H-WINX1. 
 
             IF (L-PSF-PROV-OSCAR = '390019' OR '390049' OR '390162' 
                                OR '390194' OR '390197' OR '390263') 
                     AND (L-PSF-GEO-MSA = '0240' 
                     AND L-PSF-WI-MSA = '0240' 
                     AND L-PSF-WGIDX-RECLASS = 'Y') 
                MOVE 01.0228 TO H-WINX1. 
 
             IF (L-PSF-PROV-OSCAR = '450065' OR '450072' OR '450591') 
                     AND (L-PSF-GEO-MSA = '3360' 
                     AND L-PSF-WI-MSA = '3360' 
                     AND L-PSF-WGIDX-RECLASS = 'Y') 
                MOVE 00.9388 TO H-WINX1. 
 
             IF (L-PSF-PROV-OSCAR = '470003') 
                     AND (L-PSF-GEO-MSA = '1123' 
                     AND L-PSF-WI-MSA = '1123' 
                     AND L-PSF-WGIDX-RECLASS = 'Y') 
                MOVE 01.1359 TO H-WINX1. 
 
         0220-CHNG-WAGEINDX-EXIT. 
             EXIT. 
 
      *************************************************************** 
      *    FOR FY 2001 NEW LUGAR HOSPITALS ONLY                     * 
      *************************************************************** 
 
         0225-CHNG-WAGEINDX. 
 
             IF (L-PSF-PROV-OSCAR = '010043') 
                     AND (L-PSF-WI-MSA = '1000' 
                     AND L-PSF-WGIDX-RECLASS = 'Y') 
                MOVE 00.8490 TO H-WINX1. 
 
             IF (L-PSF-PROV-OSCAR = '010072' OR '010101') 
                     AND (L-PSF-WI-MSA = '0450' 
                     AND L-PSF-WGIDX-RECLASS = 'Y') 
                MOVE 00.7871 TO H-WINX1. 
 
             IF (L-PSF-PROV-OSCAR = '100098') 
                     AND (L-PSF-WI-MSA = '8960' 
                     AND L-PSF-WGIDX-RECLASS = 'Y') 
                MOVE 00.9615 TO H-WINX1. 
 
             IF (L-PSF-PROV-OSCAR = '100232') 
                     AND (L-PSF-WI-MSA = '2900' 
                     AND L-PSF-WGIDX-RECLASS = 'Y') 
                MOVE 01.0074 TO H-WINX1. 
 
             IF (L-PSF-PROV-OSCAR = '110130') 
                     AND (L-PSF-WI-MSA = '0500' 



                     AND L-PSF-WGIDX-RECLASS = 'Y') 
                MOVE 00.9739 TO H-WINX1. 
 
             IF (L-PSF-PROV-OSCAR = '140230') 
                     AND (L-PSF-WI-MSA = '1400' 
                     AND L-PSF-WGIDX-RECLASS = 'Y') 
                MOVE 00.9069 TO H-WINX1. 
 
             IF (L-PSF-PROV-OSCAR = '230027') 
                     AND (L-PSF-WI-MSA = '3000' 
                     AND L-PSF-WGIDX-RECLASS = 'Y') 
                MOVE 01.0119 TO H-WINX1. 
 
             IF (L-PSF-PROV-OSCAR = '340071' OR '340124') 
                     AND (L-PSF-WI-MSA = '6640' 
                     AND L-PSF-WGIDX-RECLASS = 'Y') 
                MOVE 00.9506 TO H-WINX1. 
 
             IF (L-PSF-PROV-OSCAR = '390030' OR '390181' OR '390183') 
                     AND (L-PSF-WI-MSA = '6680' 
                     AND L-PSF-WGIDX-RECLASS = 'Y') 
                MOVE 00.8992 TO H-WINX1. 
 
             IF (L-PSF-PROV-OSCAR = '390201') 
                     AND (L-PSF-WI-MSA = '5640' 
                     AND L-PSF-WGIDX-RECLASS = 'Y') 
                MOVE 01.0890 TO H-WINX1. 
 
         0225-CHNG-WAGEINDX-EXIT. 
             EXIT. 
 
      *************************************************************** 
      *    CALCULATE DISCOUNT FACTOR BASED ON THE DISCOUNT          * 
      *    INDICATOR PASSED BY THE OCE: VALUE 1 - 8                 * 
      *    IF MISSING OR INVALID DISCOUNT FACTOR                    * 
      *       - SET RETURN CODE TO '38'                             * 
      *         - DISCONTINUE LINE PROCESSING                       * 
      *************************************************************** 
         0250-CALC-DISCOUNT. 
 
             IF OPPS-DISC-FACT (LN-SUB) = 1 THEN 
                MOVE 1 TO H-DISC-RATE 
             ELSE 
              IF OPPS-DISC-FACT (LN-SUB) = 2 THEN 
                 COMPUTE H-DISC-RATE = (1 + DISC-FRACTION  * 
                 (H-SRVC-UNITS - 1)) / H-SRVC-UNITS 
              ELSE 
               IF OPPS-DISC-FACT (LN-SUB) = 3 THEN 
                  COMPUTE H-DISC-RATE = TERM-PROC-DISC 
                                     / H-SRVC-UNITS 
               ELSE 
                IF OPPS-DISC-FACT (LN-SUB) = 4 THEN 
                   COMPUTE H-DISC-RATE = (1 + DISC-FRACTION) 
                                     / H-SRVC-UNITS 
                ELSE 
                 IF OPPS-DISC-FACT (LN-SUB) = 5 THEN 
                    COMPUTE H-DISC-RATE = DISC-FRACTION 



                 ELSE 
                  IF OPPS-DISC-FACT (LN-SUB) = 6 THEN 
                     COMPUTE H-DISC-RATE = (TERM-PROC-DISC * 
                        DISC-FRACTION) / H-SRVC-UNITS 
                  ELSE 
                   IF OPPS-DISC-FACT (LN-SUB) = 7 THEN 
                      COMPUTE H-DISC-RATE = (DISC-FRACTION * 
                      (1 + DISC-FRACTION) / H-SRVC-UNITS) 
                   ELSE 
                    IF OPPS-DISC-FACT (LN-SUB) = 8 THEN 
                       MOVE 2 TO H-DISC-RATE 
                    ELSE 
                     MOVE  38  TO A-RETURN-CODE (LN-SUB). 
 
         0250-CALC-DISCOUNT-EXIT. 
             EXIT. 
 
      *************************************************************** 
      *    DETERMINE THE DEDUCTIBLE SEQUENCE. DEDUCTIBLE WILL BE    * 
      *    TAKEN FROM OPPS SERVICES FIRST, THEN FROM ANY OTHER      * 
      *    TYPES OF SERVICES FROM THE CLAIM.                        * 
      *       - SET POINTERS TO THE HIGHEST RANKED PERCENTAGE       * 
      *         (LARGEST NATIONAL UNADJUSTED COINSURANCE /          * 
      *          THE APC PAYMENT)                                   * 
      *       - MOVE ALL PRICING VARIABLES TO STAGING AREA          * 
      *************************************************************** 
         0300-COIN-DEDUCT. 
 
             ADD 1 TO W-LNC-MAX. 
             SET W-LP-INDX TO W-LNC-MAX. 
             PERFORM 0350-STAGE-ENTRY 
                THRU 0350-STAGE-ENTRY-EXIT 
                   UNTIL W-LP-INDX = 1 OR 
                     H-RANK NOT < W-RANK (W-LP-INDX - 1). 
             MOVE LN-SUB TO W-LP-SUB (W-LP-INDX). 
             MOVE H-NAT-COIN TO W-NAT-COIN (W-LP-INDX). 
             MOVE H-MIN-COIN TO W-MIN-COIN (W-LP-INDX). 
             MOVE H-APC-PYMT TO W-APC-PYMT (W-LP-INDX). 
             MOVE H-WINX1    TO W-WINX1 (W-LP-INDX). 
             MOVE H-RANK     TO W-RANK (W-LP-INDX). 
             MOVE H-PPCT     TO W-PPCT (W-LP-INDX). 
             MOVE H-DISC-RATE TO W-DISC-RATE (W-LP-INDX). 
 
             IF OPPS-SRVC-IND (LN-SUB) = 'P' THEN 
                 MOVE 1 TO W-SRVC-UNITS (W-LP-INDX) 
             ELSE 
                 MOVE H-SRVC-UNITS TO W-SRVC-UNITS (W-LP-INDX). 
 
             MOVE 0 TO W-RED-COIN (W-LP-INDX). 
             PERFORM VARYING PS-SUB FROM 1 BY 1 
              UNTIL PS-SUB > L-PSF-APC-LINE-CNT 
              IF L-PSF-APC (PS-SUB) = OPPS-APC (LN-SUB) 
                COMPUTE W-RED-COIN (W-LP-INDX) ROUNDED = 
                L-PSF-RED-COIN (PS-SUB) * H-SRVC-UNITS * H-DISC-RATE 
                MOVE 25 TO A-RETURN-CODE (LN-SUB) 
                MOVE L-PSF-APC-LINE-CNT TO PS-SUB 
              END-IF 



             END-PERFORM. 
 
         0300-COIN-DEDUCT-EXIT. 
             EXIT. 
 
         0350-STAGE-ENTRY. 
 
             MOVE W-LP-ENTRY (W-LP-INDX - 1) TO 
                  W-LP-ENTRY (W-LP-INDX). 
             SET W-LP-INDX DOWN BY 1. 
 
         0350-STAGE-ENTRY-EXIT. 
             EXIT. 
 
      *************************************************************** 
      *    FIRST STEP IN DETERMINING A LINE ITEM PRICE.             * 
      *    CHECK ALL APPROPRIATE FLAGS AND INDICATORS PASSED        * 
      *    BY THE OCE.                                              * 
      *      - SET RETURN CODES TO INDICATE ERRORS OR STATUS        * 
      *        - '20' - LINE PROCESSED BUT PAYMENT = 0              * 
      *                 - BENE DEDUCTIBLE => ADJUSTED PAYMENT       * 
      *                                                             * 
      *************************************************************** 
         0400-CALCULATE. 
 
             MOVE W-LP-SUB (W-LP-INDX) TO LN-SUB. 
             IF A-RETURN-CODE (LN-SUB) >  25 
                GO TO 0400-CALCULATE-EXIT. 
 
             IF OPPS-PYMT-IND (LN-SUB) = '1' OR '5' 
                               OR '6' OR '7' OR '8' 
                PERFORM 0550-CALC-STANDARD 
                   THRU 0550-CALC-STANDARD-EXIT 
             ELSE 
                GO TO 0400-CALCULATE-EXIT. 
 
      *************************************************************** 
      *  SET STVX AND / OR GJK FLAGS                                * 
      *    - TEST LINE ITEM DATE OF SERVICE > 20010630              * 
      *    - TOTAL DRUG / DEVICE COINSURANCE                        * 
      *************************************************************** 
 
             IF (A-RETURN-CODE (LN-SUB) <  30) AND 
                (OPPS-LITEM-DOS (LN-SUB) > 20001231) 
               PERFORM 0450-ADJ-PROC-COIN 
                  THRU 0450-ADJ-PROC-COIN-EXIT 
             ELSE 
               NEXT SENTENCE. 
 
             IF A-RETURN-CODE (LN-SUB) <  30 
               COMPUTE A-TOTAL-CLM-DEDUCT = 
                       H-TOTAL-LN-DEDUCT + A-TOTAL-CLM-DEDUCT 
               COMPUTE H-TOT-PYMT = H-TOT-PYMT + H-LITEM-PYMT 
               MOVE H-LITEM-PYMT TO A-LITEM-PYMT (LN-SUB) 
               MOVE H-LITEM-REIM TO A-LITEM-REIM (LN-SUB) 
               MOVE H-TOTAL-LN-DEDUCT TO A-TOTAL-LN-DEDUCT (LN-SUB) 
               MOVE H-NAT-COIN TO A-ADJ-COIN (LN-SUB) 



               MOVE H-RED-COIN TO A-RED-COIN (LN-SUB). 
               IF H-RED-COIN > H-NAT-COIN 
                 MOVE H-NAT-COIN TO A-RED-COIN (LN-SUB). 
             MOVE ZERO TO LINE-HOLD-ITEMS. 
 
         0400-CALCULATE-EXIT. 
             EXIT. 
 
      *************************************************************** 
      *  SET STVX AND / OR GJK FLAGS                                * 
      *    - STAGE BY SERVICE INDICATOR                             * 
      *                                                             * 
      *************************************************************** 
         0450-ADJ-PROC-COIN. 
 
             MOVE OPPS-LITEM-DOS (LN-SUB) TO H-DCP-DOS. 
             IF OPPS-SRVC-IND (LN-SUB) = 'S' OR 'T' OR 'V' 
                COMPUTE H-NEW-WGNAT ROUNDED = 
                      W-NAT-COIN (W-LP-INDX) * 
                         (.6 * W-WINX1 (W-LP-INDX) + .4) 
                  IF H-NEW-WGNAT > H-IP-LIMIT 
                    MOVE H-IP-LIMIT TO H-NEW-WGNAT 
                  END-IF 
                COMPUTE H-NEW-COIN = H-LITEM-PYMT - 
                     H-TOTAL-LN-DEDUCT - H-LITEM-REIM 
                MOVE 1 TO H-DCP-CODE 
                PERFORM 0455-SEARCH-KEY 
                   THRU 0455-SEARCH-KEY-EXIT 
             ELSE 
                IF OPPS-SRVC-IND (LN-SUB) = 'G' OR 'J' OR 'K' 
                   COMPUTE H-NEW-COIN = H-LITEM-PYMT - 
                        H-TOTAL-LN-DEDUCT - H-LITEM-REIM 
                   MOVE 'Y' TO GJK-FLAG 
                   MOVE 1 TO H-DCP-CODE 
                   PERFORM 0455-SEARCH-KEY 
                      THRU 0455-SEARCH-KEY-EXIT 
                   MOVE 2 TO H-DCP-CODE 
                   ADD 1 TO W-DCP-MAX 
                   SET W-DCP-INDX TO W-DCP-MAX 
                   PERFORM 0475-STAGE-DCP-ENTRY 
                     THRU 0475-STAGE-DCP-ENTRY-EXIT 
                       UNTIL W-DCP-INDX = 1 OR 
                       H-DCP-STAGE NOT < W-DCP-STAGE (W-DCP-INDX - 1) 
                   MOVE LN-SUB TO W-DCP-SUB (W-DCP-INDX) 
                   MOVE H-DCP-STAGE TO W-DCP-STAGE (W-DCP-INDX) 
                   MOVE ZERO TO W-DCP-COIN1 (W-DCP-INDX) 
                   MOVE ZERO TO W-DCP-WGNAT (W-DCP-INDX) 
                   MOVE H-NEW-COIN TO W-DCP-COIN2 (W-DCP-INDX) 
                   MOVE OPPS-SRVC-IND (LN-SUB) TO 
                               W-DCP-SRVC-IND (W-DCP-INDX). 
 
         0450-ADJ-PROC-COIN-EXIT. 
            EXIT. 
 
         0455-SEARCH-KEY. 
 
             SET W-DCP-INDX TO 1. 



             SEARCH  W-DCP-ENTRY VARYING W-DCP-INDX 
                AT END 
                   PERFORM 0460-ADD-ENTRY 
                      THRU 0460-ADD-ENTRY-EXIT 
                WHEN W-DCP-STAGE (W-DCP-INDX) = H-DCP-STAGE 
                   PERFORM 0465-UPDATE-ENTRY 
                      THRU 0465-UPDATE-ENTRY-EXIT. 
 
         0455-SEARCH-KEY-EXIT. 
             EXIT. 
 
         0460-ADD-ENTRY. 
 
             ADD 1 TO W-DCP-MAX. 
             SET W-DCP-INDX TO W-DCP-MAX. 
             PERFORM 0475-STAGE-DCP-ENTRY 
               THRU 0475-STAGE-DCP-ENTRY-EXIT 
                UNTIL W-DCP-INDX = 1 OR 
                  H-DCP-STAGE NOT < W-DCP-STAGE (W-DCP-INDX - 1). 
             IF OPPS-SRVC-IND (LN-SUB) = 'G' OR 'J' OR 'K' 
                MOVE ZERO TO W-DCP-SUB (W-DCP-INDX) 
                MOVE H-DCP-STAGE TO W-DCP-STAGE (W-DCP-INDX) 
                MOVE ZERO TO W-DCP-COIN1 (W-DCP-INDX) 
                MOVE ZERO TO W-DCP-WGNAT (W-DCP-INDX) 
                MOVE H-NEW-COIN TO W-DCP-COIN2 (W-DCP-INDX) 
                MOVE 'X' TO W-DCP-SRVC-IND (W-DCP-INDX) 
             ELSE 
                MOVE LN-SUB TO W-DCP-SUB (W-DCP-INDX) 
                MOVE H-DCP-STAGE TO W-DCP-STAGE (W-DCP-INDX) 
                MOVE H-NEW-COIN TO W-DCP-COIN1 (W-DCP-INDX) 
                MOVE H-NEW-WGNAT TO W-DCP-WGNAT (W-DCP-INDX) 
                MOVE ZERO TO W-DCP-COIN2 (W-DCP-INDX) 
                MOVE OPPS-SRVC-IND (LN-SUB) TO 
                                  W-DCP-SRVC-IND (W-DCP-INDX). 
 
         0460-ADD-ENTRY-EXIT. 
             EXIT. 
 
         0465-UPDATE-ENTRY. 
 
             IF OPPS-SRVC-IND (LN-SUB) = 'G' OR 'J' OR 'K' 
               ADD H-NEW-COIN TO W-DCP-COIN2 (W-DCP-INDX) 
             ELSE 
               IF W-DCP-SRVC-IND (W-DCP-INDX) = 'X' 
                  PERFORM 0485-REPLACE-TYPE1 
                     THRU 0485-REPLACE-TYPE1-EXIT 
               ELSE 
                  PERFORM 0480-RANK-COIN 
                     THRU 0480-RANK-COIN-EXIT. 
 
         0465-UPDATE-ENTRY-EXIT. 
             EXIT. 
 
         0475-STAGE-DCP-ENTRY. 
 
             MOVE W-DCP-ENTRY (W-DCP-INDX - 1) TO 
                  W-DCP-ENTRY (W-DCP-INDX). 



             SET W-DCP-INDX DOWN BY 1. 
 
         0475-STAGE-DCP-ENTRY-EXIT. 
             EXIT. 
 
         0480-RANK-COIN. 
 
             IF H-NEW-WGNAT > W-DCP-WGNAT (W-DCP-INDX) 
               MOVE LN-SUB TO W-DCP-SUB (W-DCP-INDX) 
               MOVE H-NEW-COIN TO W-DCP-COIN1 (W-DCP-INDX) 
               MOVE H-NEW-WGNAT TO W-DCP-WGNAT (W-DCP-INDX) 
               MOVE OPPS-SRVC-IND (LN-SUB) 
                           TO W-DCP-SRVC-IND (W-DCP-INDX). 
 
         0480-RANK-COIN-EXIT. 
             EXIT. 
 
         0485-REPLACE-TYPE1. 
 
             MOVE LN-SUB TO W-DCP-SUB (W-DCP-INDX) 
             MOVE H-NEW-COIN TO W-DCP-COIN1 (W-DCP-INDX) 
             MOVE H-NEW-WGNAT TO W-DCP-WGNAT (W-DCP-INDX) 
             MOVE OPPS-SRVC-IND (LN-SUB) 
                         TO W-DCP-SRVC-IND (W-DCP-INDX). 
 
         0485-REPLACE-TYPE1-EXIT. 
             EXIT. 
 
      *************************************************************** 
      * 1. APPLY DEDUCTIBLE TO HIGHEST NATIONAL COINSURANCE AMOUNT  * 
      *     - DESCENDING UNTIL DEDUCTIBLE = 0.                      * 
      * 2. CALCULATE THE STANDARD LINE PRICE                        * 
      *    (APC PAYMENT * WAGE INDEX * SERVICE UNITS                * 
      *          * DISCOUNT FACTOR)                                 * 
      *     - WAGE ADJUST 60% OF THE APC PAYMENT ONLY               * 
      * 3. ADD LINE PRICE TO OUTLIER HOLD AREA.                     * 
      *                                                             * 
      * 4. CALCULATE THE LINE PRICE FOR DESIGNATED DEVICES          * 
      *       AND DRUGS                                             * 
      *************************************************************** 
         0550-CALC-STANDARD. 
 
             COMPUTE H-MAX-COIN ROUNDED = (H-IP-LIMIT * 
               W-SRVC-UNITS (W-LP-INDX) * W-DISC-RATE (W-LP-INDX)) 
             IF OPPS-SRVC-IND (LN-SUB) = 'S' OR 'V' 
                        OR 'T' OR 'P' OR 'X' THEN 
                COMPUTE H-LITEM-PYMT ROUNDED = 
                    (((W-APC-PYMT (W-LP-INDX) * .60) * 
                            W-WINX1 (W-LP-INDX)) 
                                + (W-APC-PYMT (W-LP-INDX) * .40)) * 
                  W-SRVC-UNITS (W-LP-INDX) * W-DISC-RATE (W-LP-INDX) 
                  PERFORM 0560-CALC-BENE-DEDUCT 
                     THRU 0560-CALC-BENE-DEDUCT-EXIT 
             ELSE 
               IF OPPS-SRVC-IND (LN-SUB) = 'H' THEN 
                 IF OPPS-PYMT-IND (LN-SUB) = '6' THEN 
                   PERFORM 0555-CALC-H-STANDARD 



                      THRU 0555-CALC-H-STANDARD-EXIT 
                   PERFORM 0560-CALC-BENE-DEDUCT 
                      THRU 0560-CALC-BENE-DEDUCT-EXIT 
                 ELSE 
                   MOVE  44  TO A-RETURN-CODE (LN-SUB) 
                   GO TO 0550-CALC-STANDARD-EXIT 
                 END-IF 
               ELSE 
                IF OPPS-SRVC-IND (LN-SUB) = 'G' OR 'J' OR 'K' THEN 
                 IF OPPS-PYMT-IND (LN-SUB) = '1' OR '5' OR '7' THEN 
                   COMPUTE H-LITEM-PYMT ROUNDED = 
                 (W-APC-PYMT (W-LP-INDX) * W-SRVC-UNITS (W-LP-INDX)) 
                             * W-DISC-RATE (W-LP-INDX) 
                  PERFORM 0560-CALC-BENE-DEDUCT 
                     THRU 0560-CALC-BENE-DEDUCT-EXIT 
                 ELSE 
                  MOVE  41  TO A-RETURN-CODE (LN-SUB) 
                  GO TO 0550-CALC-STANDARD-EXIT 
                 END-IF 
                END-IF. 
 
             IF H-LITEM-PYMT > 0 
                COMPUTE H-LITEM-REIM ROUNDED = 
                (H-LITEM-PYMT - H-TOTAL-LN-DEDUCT) * 
                                     W-PPCT (W-LP-INDX) 
                COMPUTE H-NAT-COIN = H-LITEM-PYMT  - 
                                  H-TOTAL-LN-DEDUCT - H-LITEM-REIM. 
             MOVE W-MIN-COIN (W-LP-INDX) TO  H-MIN-COIN. 
             IF H-MIN-COIN > 0 
               IF OPPS-SRVC-IND (LN-SUB) = 'G' OR 'H' OR 'J' OR 'K' 
                 COMPUTE H-MIN-COIN ROUNDED = H-MIN-COIN * 
                  W-SRVC-UNITS (W-LP-INDX) * W-DISC-RATE (W-LP-INDX) 
               ELSE 
                 IF OPPS-APC (LN-SUB) = '0158' OR '0159' 
                   COMPUTE H-MIN-COIN ROUNDED = H-LITEM-PYMT * .25 
                 ELSE 
                   COMPUTE H-MIN-COIN ROUNDED = H-LITEM-PYMT * .2 
                 END-IF 
               END-IF 
             END-IF. 
 
             MOVE W-RED-COIN (W-LP-INDX) TO H-RED-COIN. 
             IF (H-RED-COIN  > 0 AND < H-MIN-COIN) 
                  MOVE H-MIN-COIN TO H-RED-COIN 
             ELSE 
                IF H-RED-COIN < H-NAT-COIN AND > H-MIN-COIN 
                   AND > H-MAX-COIN 
                   COMPUTE H-LITEM-REIM = H-LITEM-REIM + 
                                         (H-RED-COIN - H-MAX-COIN) 
                END-IF 
             END-IF. 
 
             IF H-NAT-COIN > H-MAX-COIN AND H-RED-COIN = 0 THEN 
                COMPUTE H-LITEM-REIM = H-LITEM-REIM + 
                                            (H-NAT-COIN - H-MAX-COIN) 
                MOVE H-MAX-COIN TO H-NAT-COIN. 
 



         0550-CALC-STANDARD-EXIT. 
             EXIT. 
 
         0555-CALC-H-STANDARD. 
 
             MOVE OPPS-SUB-CHRG (LN-SUB) TO H-SUB-CHRG 
             IF L-SERVICE-FROM-DATE > 20001231 
                COMPUTE T-LITEM-PYMT ROUNDED = 
               (H-SUB-CHRG * (L-PSF-OPCOST-RATIO * .981956)) - 
               (((W-APC-PYMT (W-LP-INDX) * .60) * 
                       W-WINX1 (W-LP-INDX)) 
                       + (W-APC-PYMT (W-LP-INDX) * .40)) 
             ELSE 
                COMPUTE T-LITEM-PYMT ROUNDED = 
                (H-SUB-CHRG * L-PSF-OPCOST-RATIO) - 
                (((W-APC-PYMT (W-LP-INDX) * .60) * 
                        W-WINX1 (W-LP-INDX)) 
                        + (W-APC-PYMT (W-LP-INDX) * .40)). 
 
                IF T-LITEM-PYMT < 0 THEN 
                   MOVE 0 TO H-LITEM-PYMT 
                ELSE 
                   MOVE T-LITEM-PYMT TO H-LITEM-PYMT. 
 
         0555-CALC-H-STANDARD-EXIT. 
             EXIT. 
 
         0560-CALC-BENE-DEDUCT. 
 
             IF OPPS-PYMT-ADJ-FLAG (LN-SUB) = '4' 
                GO TO 0560-CALC-BENE-DEDUCT-EXIT. 
             IF H-BENE-DEDUCT > 0 THEN 
               IF H-BENE-DEDUCT <= H-LITEM-PYMT THEN 
                 MOVE H-BENE-DEDUCT TO H-TOTAL-LN-DEDUCT 
                 MOVE 0 TO H-BENE-DEDUCT 
               ELSE 
                  COMPUTE H-BENE-DEDUCT = 
                          H-BENE-DEDUCT - H-LITEM-PYMT 
                  MOVE H-LITEM-PYMT TO H-TOTAL-LN-DEDUCT 
                  MOVE  20  TO A-RETURN-CODE (LN-SUB) 
               END-IF 
             END-IF. 
 
         0560-CALC-BENE-DEDUCT-EXIT. 
             EXIT. 
 
         0800-ADJ-STV-REIM. 
 
             IF W-DCP-CODE (W-DCP-INDX) = 1 
                PERFORM 0810-PROCESS-TYPE1 
                   THRU 0810-PROCESS-TYPE1-EXIT 
             ELSE 
                PERFORM 0840-PROCESS-TYPE2 
                   THRU 0840-PROCESS-TYPE2-EXIT. 
 
         0800-ADJ-STV-REIM-EXIT. 
             EXIT. 



 
         0810-PROCESS-TYPE1. 
 
             IF  W-DCP-COIN2 (W-DCP-INDX) > 0 
                MOVE W-DCP-DOS (W-DCP-INDX) TO H-DCP-DOS 
                MOVE W-DCP-WGNAT (W-DCP-INDX) TO H-TOTAL 
                COMPUTE H-RATIO = 
                   (H-IP-LIMIT - W-DCP-WGNAT (W-DCP-INDX)) / 
                                  W-DCP-COIN2 (W-DCP-INDX) 
                IF H-RATIO < 0 
                    MOVE 0 TO H-RATIO. 
                IF H-RATIO > 1 
                    MOVE 1 TO H-RATIO. 
 
         0810-PROCESS-TYPE1-EXIT. 
             EXIT. 
 
         0840-PROCESS-TYPE2. 
 
             IF W-DCP-DOS (W-DCP-INDX) >= 20010701 
                IF W-DCP-DOS (W-DCP-INDX) =  H-DCP-DOS 
                   MOVE W-DCP-SUB (W-DCP-INDX) TO LN-SUB 
                   COMPUTE H-SHIFT = 
                       W-DCP-COIN2 (W-DCP-INDX) * (1 - H-RATIO) 
                   COMPUTE H-TOTAL = 
                        A-ADJ-COIN (LN-SUB) + H-TOTAL - H-SHIFT 
                   IF H-TOTAL > H-IP-LIMIT 
                      COMPUTE H-SHIFT = H-SHIFT + H-TOTAL 
                            - H-IP-LIMIT 
                   END-IF 
                   COMPUTE A-ADJ-COIN (LN-SUB) = 
                        A-ADJ-COIN (LN-SUB) - H-SHIFT 
                   COMPUTE A-LITEM-REIM (LN-SUB) = 
                      A-LITEM-REIM (LN-SUB) + H-SHIFT 
                   MOVE 22 TO A-RETURN-CODE (LN-SUB) 
                END-IF 
             ELSE 
                IF W-DCP-DOS (W-DCP-INDX) >= 20010101 
                  IF W-DCP-COIN2 (W-DCP-INDX) > H-IP-LIMIT 
                     MOVE W-DCP-SUB (W-DCP-INDX) TO LN-SUB 
                     COMPUTE H-SHIFT = 
                         W-DCP-COIN2 (W-DCP-INDX) - H-IP-LIMIT 
                     COMPUTE A-ADJ-COIN (LN-SUB) = 
                          A-ADJ-COIN (LN-SUB) - H-SHIFT 
                     COMPUTE A-LITEM-REIM (LN-SUB) = 
                        A-LITEM-REIM (LN-SUB) + H-SHIFT 
                     MOVE 22 TO A-RETURN-CODE (LN-SUB) 
                  END-IF 
                END-IF 
             END-IF. 
 
 
         0840-PROCESS-TYPE2-EXIT. 
             EXIT. 
 
      *************************************************************** 
      * 1. CALCULATE CLAIM LEVEL OUTLIER AMOUNT                     * 



      *************************************************************** 
         0900-END-PRICE-RTN. 
 
             MOVE H-TOT-CHRG TO A-TOT-CLM-CHRG. 
             MOVE H-TOT-PYMT TO A-TOT-CLM-PYMT. 
             IF A-TOT-CLM-PYMT > 0 
                PERFORM 0910-CALC-OUTLIER 
                   THRU 0910-CALC-OUTLIER-EXIT. 
             IF H-OUTLIER-PYMT > 0 
                MOVE H-OUTLIER-PYMT TO A-OUTLIER-PYMT. 
 
         0900-END-PRICE-RTN-EXIT. 
             EXIT. 
 
         0910-CALC-OUTLIER. 
 
             IF L-SERVICE-FROM-DATE > 20001231 
              COMPUTE H-OUTLIER-PYMT ROUNDED = 
                ((H-TOT-CHRG * (L-PSF-OPCOST-RATIO * .981956)) 
                      - (2.5 * H-TOT-PYMT)) * .75 
             ELSE 
              COMPUTE H-OUTLIER-PYMT ROUNDED = 
                ((H-TOT-CHRG * L-PSF-OPCOST-RATIO) 
                      - (2.5 * H-TOT-PYMT)) * .75. 
 
         0910-CALC-OUTLIER-EXIT. 
             EXIT. 
� 


